

```

gap> g:= SymmetricGroup( 4 );
Sym( [ 1 .. 4 ] )
gap> tbl:= CharacterTable( g );; HasIrr( tbl );
i5 : betti(t,Weights=>{1,0})
false
      0 1 2 3 4
o5 = total: 1 4 13 14 4
      0: 1 . . . .
      1: . 2 2 4 2
      2: . 2 5 6 .
      3: . . 4 . 2
      4: . . . 4 .
      5: . . 2 . .
gap> tblmod2:= CharacterTable( tbl, 2 );
BrauerTable( Sym( [ 1 .. 4 ] ), 2 )
gap> tblmod2 = CharacterTable( tbl, 2 );
true
gap> tblmod2 = BrauerTable( tbl, 2 );
true
o5 : BrauerTable( Sym( [ 1 .. 4 ] ), 2 )
i6 : betti(t,Weights=>{0,1})
      0 1 2 3 4
o6 = total: 1 4 13 14 4
      0: 1 . . . .
      1: . 2 2 4 2
      2: . 2 5 6 .
      3: . . 4 . 2
      4: . . . 4 .
      5: . . 2 . .
gap> libtbl:= CharacterTable( "M" );
CharacterTable( "M" )
gap> CharacterTableRegular( libtbl, 2 );
BrauerTable( "M", 2 )
gap> BrauerTable( libtbl, 2 );
fail
gap> CharacterTable( "Symmetric", 4 );
CharacterTable( "Sym(4)" )
i7 : t1 = betti(t,Weights=>{1,1})
gap> ComputedBrauerTables( tbl );
[ , BrauerTable( Sym( [ 1 .. 4 ] ), 2 ) ]
      0 1 2 3 4
o7 = total: 1 4 13 14 4
      0: 1 . . . .
      1: . . . . .
      2: . . . . .
      3: . 2 . . .
      4: . . . . .
      5: . 2 . . .
      6: . . 1 . .
      7: . . 8 6 .
      8: . . 4 8 4
      ring r1 = 32003,(x,y,z),ds;
      int a,b,c,t=11,5,3,0;
      poly f = x^a+y^b+z^(3*c)+x^(c+2)*y^(c-1)+x^(
      x^(c-2)*y^c*(y^2+t*x)^2;
      option(noprot);
      timer=1;
      ring r2 = 32003,(x,y,z),dp;
      poly f=imap(r1,f);
      ideal j=jacob(f);
      vdim(std(j));
      ==> 536
      vdim(std(j+f));
      ==> 195
      timer=0; // reset timer
o7 : BettiTally
o8 : peek t1
o8 = BettiTally{(0, {0, 0}, 0) => 1 }
      (1, {2, 2}, 4) => 2
      (1, {3, 3}, 6) => 2
      (2, {3, 7}, 10) => 2
      (2, {4, 4}, 8) => 1
      (2, {4, 5}, 9) => 4
      (2, {5, 4}, 9) => 4
      (2, {7, 3}, 10) => 2
      (3, {4, 7}, 11) => 4
      (3, {5, 5}, 10) => 6
      (3, {7, 3}, 10) => 2
      (4, {5, 7}, 12) => 2
      (4, {7, 5}, 12) => 2

```

# Journal of Software for Algebra and Geometry

The virtual resolutions package for Macaulay2

AYAH ALMOUSA, JULIETTE BRUCE, MICHAEL LOPER AND MAHRUD SAYRAFI



## The virtual resolutions package for Macaulay2

AYAH ALMOUSA, JULIETTE BRUCE, MICHAEL LOPER AND MAHRUD SAYRAFI

**ABSTRACT:** We introduce the `VirtualResolutions` package for the computer algebra system *Macaulay2*. This package has tools to construct, display, and study virtual resolutions for products of projective spaces. The package also has tools for generating curves in  $\mathbb{P}^1 \times \mathbb{P}^2$ , providing sources of interesting virtual resolutions.

**1. INTRODUCTION.** Recently, Berkesch, Erman, and Smith [Berkesch et al. 2017] introduced the notion of virtual resolutions for subvarieties of smooth projective toric varieties as an analogue to minimal graded free resolutions for subvarieties of projective space. A virtual resolution is a complex of finitely generated graded free modules over the Cox ring of a smooth projective toric variety that becomes exact upon passing to the corresponding complex of coherent sheaves.

While graded minimal free resolutions are useful for studying quasicoherent sheaves on projective spaces, when working over a product of projective spaces or, more generally, over smooth projective toric varieties, they are often long and cumbersome to compute. By allowing a limited amount of homology, virtual resolutions offer a more flexible alternative for studying toric subvarieties when compared to graded minimal free resolutions. This article introduces the `VirtualResolutions` package for [Macaulay2], which provides tools for constructing and studying virtual resolutions over products of projective spaces.

The `VirtualResolutions` project began in 2018 at the *Macaulay2 Workshop at the University of Wisconsin – Madison*, building on previous work of Christine Berkesch, David Eisenbud, Daniel Erman, and Gregory G. Smith. Along with them, the authors of this paper developed the `VirtualResolutions` package to generate examples of virtual resolutions in products of projective spaces. In particular, the package provides tools to generate examples of virtual resolutions and to check whether chain complexes are virtual resolutions. These methods are introduced and demonstrated in Section 2.

In Section 3, we concentrate on virtual resolutions arising from curves in  $\mathbb{P}^1 \times \mathbb{P}^2$ . This is the simplest case after points in a product of projective spaces. We demonstrate several methods for finding the

---

Almoussa was partially supported by the NSF GRFP under Grant No. DGE-1650441. Bruce was partially supported by the NSF GRFP under Grant No. DGE-1256259 and NSF grant DMS-1502553. Loper was partially supported by the NSF RTG grant DMS-1745638.

*MSC2010:* 13D02, 14M25.

*Keywords:* virtual resolution, products of projective spaces, toric varieties, free resolutions.

`VirtualResolutions` version 1.2

defining ideals of curves in the Cox ring of  $\mathbb{P}^1 \times \mathbb{P}^2$ , including monomial curves, rational curves, and curves from  $\mathbb{P}^3$ .

**2. CONSTRUCTING VIRTUAL RESOLUTIONS.** If  $X$  is a smooth projective toric variety, we denote its  $\text{Pic}(X)$ -graded Cox ring, as defined in [Cox 1995], by  $\text{Cox}(X)$  (see also Section 5.2 of [Cox et al. 2011]). Denote the associated irrelevant ideal of  $X$  by  $B \subset \text{Cox}(X)$ . A virtual resolution of a  $\text{Pic}(X)$ -graded module over  $\text{Cox}(X)$  is defined as follows:

**Definition 2.1** [Berkesch et al. 2017, Definition 1.1]. A *virtual resolution* of a  $\text{Pic}(X)$ -graded  $\text{Cox}(X)$ -module  $M$  is a chain complex of  $\text{Pic}(X)$ -graded free  $\text{Cox}(X)$ -modules

$$\mathbf{F}_\bullet := [F_0 \longleftarrow F_1 \longleftarrow F_2 \longleftarrow \cdots]$$

such that the corresponding complex of  $\mathcal{O}_X$ -modules is a locally free resolution of the sheaf  $\tilde{M}$ .

One can rephrase the definition of a virtual resolution in a way that is more practical for computations: a complex  $\mathbf{F}_\bullet$  of  $\text{Pic}(X)$ -graded free  $\text{Cox}(X)$ -modules is a virtual resolution of  $M$  if and only if the following two conditions are satisfied:

- (1) The  $B$ -saturation of  $H_0(\mathbf{F}_\bullet)$  is isomorphic to the  $B$ -saturation of  $M$ .
- (2) The homology module  $H_i(\mathbf{F}_\bullet)$  is supported only on  $B$  for all  $i > 0$ .

For the remainder of this paper, we will focus on the case when  $X$  is a product of projective spaces, since [Berkesch et al. 2017] demonstrates several ways to construct virtual resolutions in this case. We let  $\mathbf{n} = (n_1, n_2, \dots, n_r) \in (\mathbb{Z}_{\geq 0})^r$  and  $\mathbb{P}^{\mathbf{n}} = \mathbb{P}^{n_1} \times \mathbb{P}^{n_2} \times \cdots \times \mathbb{P}^{n_r}$ . Further, we write  $S$  for  $\text{Cox}(\mathbb{P}^{\mathbf{n}})$ , which is graded by  $\text{Pic}(\mathbb{P}^{\mathbf{n}}) \cong \mathbb{Z}^r$ .

Given this setup, Berkesch, Erman, and Smith show that it is possible to construct a virtual resolution of a  $\mathbb{Z}^r$ -graded  $S$ -module  $M$  from the graded minimal free resolution of  $M$  and  $\mathbf{d} \in \mathbb{Z}^r$  an element of the multigraded Castelnuovo–Mumford regularity as introduced in [Maclagan and Smith 2004]. They call a virtual resolution constructed in this manner the *virtual resolution of the pair*  $(M, \mathbf{d})$ .

**Theorem 2.2** [Berkesch et al. 2017, Theorem 1.3]. *Let  $M$  be a finitely generated  $\mathbb{Z}^r$ -graded  $B$ -saturated  $S$ -module that is  $\mathbf{d}$ -regular. If  $G$  is the free subcomplex of a minimal free resolution of  $M$  consisting of all summands generated in degrees at most  $\mathbf{d} + \mathbf{n}$ , then  $G$  is a virtual resolution of  $M$ .*

Notice that the virtual resolution of the pair  $(M, \mathbf{d})$  is a subcomplex of the minimal free resolution, so its length is at most the projective dimension of  $M$ . Also note that we are justified in saying “the” virtual resolution of the pair  $(M, \mathbf{d})$  because this complex is unique up to isomorphism.

**Example 2.3.** Consider three points  $([1 : 1], [1 : 4])$ ,  $([1 : 2], [1 : 5])$ , and  $([1 : 3], [1 : 6])$  in  $\mathbb{P}^1 \times \mathbb{P}^1$ . Using *Macaulay2*, we may compute the  $B$ -saturated ideal defining these three points.

```
i1 : needsPackage "VirtualResolutions";
i2 : X = toricProjectiveSpace(1) ** toricProjectiveSpace(1);
i3 : S = ring X; B = ideal X;
o3 : Ideal of S
```

```

i4 : J = saturate(intersect(
      ideal(x_1 - x_0, x_3 - 4*x_2),
      ideal(x_1 - 2*x_0, x_3 - 5*x_2),
      ideal(x_1 - 3*x_0, x_3 - 6*x_2)), B);
o4 : Ideal of S
    
```

One can show the  $\mathbb{Z}^2$ -graded minimal free resolution of  $S/J$  is the following complex.

$$\begin{array}{ccccccc}
 & & S(0, -3) & & & & \\
 & & \oplus & & & & \\
 & & S(-1, -2) & & S(-1, -3)^2 & & \\
 & & \oplus & & \oplus & & S(-2, -3) \\
 S \longleftarrow & S(-2, -1) & \longleftarrow & S(-2, -2)^2 & \longleftarrow & \oplus & \longleftarrow 0 \\
 & \oplus & & \oplus & & S(-3, -2) & \\
 & S(-3, 0) & & S(-3, -1)^2 & & & \\
 & \oplus & & & & & \\
 & S(-1, -1) & & & & & 
 \end{array}$$

We may view the multigraded Betti table for the above graded minimal free resolution.

```

i5 : minres = res J;
i6 : multigraded betti minres
      0          1          2          3
o6 = 0: 1          .          .          .
      2: .          ab         .          .
      3: . a3+a2b+ab2+b3      .          .
      4: .          . 2a3b+2a2b2+2ab3 .
      5: .          .          . a3b2+a2b3
o6 : MultigradedBettiTally
    
```

In order to compute a virtual resolution of  $S/J$ , we find an element of the multigraded Castelnuovo–Mumford regularity of the module [Maclagan and Smith 2004, Definition 4.1]. The minimal elements in the multigraded regularity of  $S/J$  can be computed via the `multigradedRegularity` command.

```

i7 : multigradedRegularity(X, S^1/J)
o7 = {{0, 2}, {1, 1}, {2, 0}}
o7 : List
    
```

This computation relies on the fact that a  $B$ -saturated  $S$ -module  $M$  over a product of projective spaces  $\mathbb{P}^n$  is  $\mathbf{d}$ -regular provided that the following conditions are satisfied:

- (1) The Hilbert function  $H(M, \mathbf{d})$  agrees with the Hilbert polynomial  $P_M(\mathbf{d})$ .
- (2)  $H^i(\mathbb{P}^n, \tilde{M}(\mathbf{a})) = 0$  for all  $i > 0$  and twists  $\mathbf{a}$  such that  $d_j - c_j \leq a_j$  with  $0 \leq c_j$  and  $\sum_j c_j = i$ .

In particular, the function `multigradedRegularity` utilizes the *Macaulay2* package `TateOnProducts` [Erman et al. 2019] to compute the sheaf cohomology of twists of  $\tilde{M}$ , which allows one to find elements of the multigraded regularity of the sheaf [Eisenbud et al. 2015, Proposition 3.11].

The function `virtualOfPair` implements Theorem 2.2 to compute the virtual resolution of a pair. We call the function below in order to compute the virtual resolution of the pair  $(S/J, (2, 0))$ . Note that

since we must remove all twists generated in degrees not less than or equal to  $n + d$  from the graded minimal free resolution, we input the element  $(3, 1) = (1, 1) + (2, 0)$  in `virtualOfPair`.

```
i8 : vres = virtualOfPair(res J, {{3, 1}})
```

```
o8 = S 1 <-- S 3 <-- S 2 <-- 0
```

```
i9 : multigraded betti vres
```

```
o9 = 0: 1 0 1 2
```

```
2: . ab .
```

```
3: . a3+a2b .
```

```
4: . . 2a3b
```

```
o9 : MultigradedBettiTally
```

The above virtual resolution of the pair  $(S/J, (2, 0))$  is shorter and thinner than the graded minimal free resolution of  $S/J$ .

When a minimal free resolution is already known or provided as input, `virtualOfPair` takes the appropriate subcomplex of that resolution. Otherwise, `virtualOfPair` computes a virtual resolution by using the induced Schreyer orders to iteratively compute syzygies in the desired degrees. Note that while the resulting virtual resolutions from the two strategies may have different differentials, the chain complexes are unique up to isomorphism.

Continuing the example of three points in  $\mathbb{P}^1 \times \mathbb{P}^1$ , we again use `virtualOfPair` to compute the virtual resolution of  $S/J$ . In this case, *Macaulay2* does not have the minimal free resolution cached, so `virtualOfPair` uses Schreyer's method to obtain a virtual resolution isomorphic to the one above.

```
i10 : multigraded betti virtualOfPair(S^1/J, {{3, 1}})
```

```
o10 = 0: 1 0 1 2
```

```
2: . ab .
```

```
3: . a3+a2b .
```

```
4: . . 2a3b
```

```
o10 : MultigradedBettiTally
```

**Remark 2.4.** In larger experiments, Schreyer's method is significantly more efficient than first computing a minimal free resolution. For instance, finding a virtual resolution for the saturated ideal of five points in  $(\mathbb{P}^1)^4$  using Schreyer's method is two orders of magnitude faster. These time savings grow even more dramatically as the number of points or number of factors of  $\mathbb{P}^1$  are increased.

We may check that `vres` is indeed a virtual resolution by using the `isVirtual` function.

```
i11 : isVirtual(B, vres)
```

```
o11 = true
```

By default, `isVirtual` checks whether the homology of the given chain complex is supported only on the irrelevant ideal. More specifically, `isVirtual` checks whether the annihilator of the homology of the given chain complex saturates to the entire ring.

**Remark 2.5.** Computing the saturation of an ideal is a critical but computationally costly step in many aspects of studying virtual resolutions. Therefore, our package utilizes new saturation methods which are under development by Michael Stillman for future release in *Macaulay2*. These methods are currently stored in the auxiliary file `Colon.m2`, but will be removed once they have been released separately.

As an alternative approach, we also implement the method presented in Theorem 1.3 of [Loper 2019] for checking whether a complex is virtual. This is done by setting `Strategy => Determinantal`. In this case, two conditions are checked: one involving the ranks of the maps in the chain complex, and the other involving the depths of the  $B$ -saturated ideals of minors of the maps. Typically the default strategy is faster than the determinantal strategy, as the determinantal strategy must compute the ideal of minors for each map in the chain complex, which is generally computationally expensive.

Another way to generate virtual resolutions is by using the following theorem of [Berkesch et al. 2017], which provides a method for producing virtual resolutions of ideals defining zero-dimensional subschemes of  $\mathbb{P}^n$ .

**Theorem 2.6** [Berkesch et al. 2017, Theorem 4.1]. *If  $Z \subset \mathbb{P}^n$  is a zero-dimensional scheme and  $I$  is the corresponding  $B$ -saturated  $S$ -ideal, then there exists  $\mathbf{a} \in \mathbb{N}^r$  with  $a_r = 0$  such that the minimal free resolution of  $S/(I \cap B^{\mathbf{a}})$  has length equal to  $|\mathbf{n}| = \dim \mathbb{P}^n$ . Moreover, any  $\mathbf{a} \in \mathbb{N}^r$  with  $a_r = 0$  and other entries sufficiently positive yields such a virtual resolution of  $S/I$ .*

The function `resolveViaFatPoint` computes  $B^{\mathbf{a}} = \bigcap_{i=1}^r (x_{i,0}, x_{i,1}, \dots, x_{i,n_i})^{a_i}$ , intersects it with the input ideal  $J$ , and computes the minimal free resolution of  $S/(J \cap B^{\mathbf{a}})$ .

**Example 2.7.** Again consider the ideal of three points in  $\mathbb{P}^1 \times \mathbb{P}^1$  as in Example 2.3. We use the function `resolveViaFatPoint` to compute a virtual resolution of  $J$ .

```
i12 : C = resolveViaFatPoint(J, B, {2, 0})

      1      4      3
o12 = S <-- S <-- S <-- 0
      0      1      2      3

o12 : ChainComplex
i13 : multigraded betti C

      0      1      2
o13 = 0: 1      .      .
      3: . a3+3a2b .
      4: .      . 3a3b
o13 : MultigradedBettiTally
i14 : isVirtual(B, C)
o14 = true
```

Note that the virtual resolutions obtained by `resolveViaFatPoint` may not be a subcomplex of the minimal resolution of  $M$  (as demonstrated by Examples 2.3 and 2.7), and thus need not occur as the virtual resolution of  $(M, \mathbf{d})$  for a given regularity  $\mathbf{d}$ . Thus, `resolveViaFatPoint` is a useful method for producing new and interesting examples of virtual resolutions.

**3. CONSTRUCTING CURVES IN  $\mathbb{P}^1 \times \mathbb{P}^2$ .** One source of potentially interesting virtual resolutions comes from studying curves in  $\mathbb{P}^1 \times \mathbb{P}^2$ . With this in mind, the `VirtualResolutions` package contains functions for constructing a limited class of curves in  $\mathbb{P}^1 \times \mathbb{P}^2$ , which we believe may be of interest to other researchers. In particular, there are functions for generating monomial curves, rational curves, and curves arising from curves in  $\mathbb{P}^3$ .

The main function in this direction is `curveFromP3toP1P2`. Given the defining ideal of a curve in  $\mathbb{P}^3$ , this function returns the defining ideal of a curve in  $\mathbb{P}^1 \times \mathbb{P}^2$  constructed in the following way: given projections  $\pi_1 : \mathbb{P}^3 \dashrightarrow \mathbb{P}^1$  and  $\pi_2 : \mathbb{P}^3 \dashrightarrow \mathbb{P}^2$ , there is an induced rational map

$$\psi : \mathbb{P}^3 \dashrightarrow \mathbb{P}^1 \times \mathbb{P}^2,$$

and this is the map under which we are computing the image of our curve  $C$ .

The projections that `curveFromP3toP1P2` uses are fixed. In particular, it uses the coordinate projections

$$\pi_1([z_0 : z_1 : z_2 : z_3]) = [z_0 : z_1] \quad \text{and} \quad \pi_2([z_0 : z_1 : z_2 : z_3]) = [z_1 : z_2 : z_3].$$

As one might wish to preserve the degree of the curve  $C \subset \mathbb{P}^1 \times \mathbb{P}^2$ , the function `curveFromP3toP1P2` has an option called `PreserveDegree`. When this option is set to `true`, the function will return an error if the given curve intersects the base locus of these projections.

**Example 3.1.** The code below uses `curveFromP3toP1P2` to construct a curve in  $\mathbb{P}^1 \times \mathbb{P}^2$  from the defining ideal of the twisted cubic  $C \subset \mathbb{P}^3$ .

```
i15 : R = ZZ/101[z_0, z_1, z_2, z_3];
i16 : I = ideal(z_0*z_2-z_1^2, z_1*z_3-z_2^2, z_0*z_3-z_1*z_2);
o16 : Ideal of R
i17 : J = curveFromP3toP1P2 I
      2
o17 = ideal (x1,12 - x1,0x1,2, -x0,1x1,1 + x0,0x1,2, -x0,1x1,0 + x0,0x1,1)
      ZZ
o17 : Ideal of ---[x0,0, x0,1, x1,0, x1,1, x1,2]
      101
```

We can check that the ideal  $J$  defines a curve in  $\mathbb{P}^1 \times \mathbb{P}^2$  by computing its dimension. Note that since  $J$  is an ideal in the Cox ring of  $\mathbb{P}^1 \times \mathbb{P}^2$ , we expect  $J$  to be three-dimensional, since in general the dimension of a subscheme defined by a  $\text{Pic}(X)$ -graded ideal  $J \subset \text{Cox}(X)$  is given by  $\dim J - \text{rank Pic}(X)$ .

```
i18 : dim J
o18 = 3
```

The function `randomCurveP1P2` constructs a random curve in  $\mathbb{P}^1 \times \mathbb{P}^2$  by calling the `curve` function from the package `SpaceCurves` [Zhang 2018] to randomly generate a curve  $C \subset \mathbb{P}^3$  of specified degree and genus; see [Zhang 2018, Section 2] for a detailed discussion of the inner workings of `curve`. The function `randomCurveP1P2` then constructs a curve  $C' \subset \mathbb{P}^1 \times \mathbb{P}^2$  from  $C$  using `curveFromP3toP1P2`.

**Example 3.2.** Using the `randomCurveP1P2` function we produce a random curve in  $\mathbb{P}^1 \times \mathbb{P}^2$  coming from a curve of degree 7 and genus 3 in  $\mathbb{P}^3$ . Additionally, we check that the resulting ideal defines a curve.

```
i19 : I = randomCurveP1P2(7, 3);
      ZZ
o19 : Ideal of ---[x_0, x_1, x_2, x_3, x_4]
      101 0,0 0,1 1,0 1,1 1,2
i20 : S = ring I;
i21 : dim I
o21 = 3
```

Using the `multigradedRegularity` function, one can see that  $(2, 3)$  is a minimal element in the multi-graded regularity of this curve. Note that we first saturate the ideal defining our curve; this is necessary to ensure that the output of `multigradedRegularity` is correct.

```
i22 : B = intersect(ideal(x_(0,0), x_(0,1)), ideal(x_(1,0), x_(1,1), x_(1,2)));
o22 : Ideal of S
i23 : J = saturate(I, B);
o23 : Ideal of S
i24 : multigradedRegularity(S, S^1/J)
o24 = {{1, 4}, {2, 3}}
o24 : List
```

Finally, we can use this element of the multigraded regularity to compute a virtual resolution of  $S/J$ .

```
i25 : minres = res J;
i26 : vres = virtualOfPair(J, {{3, 5}})

o26 = S  $\leftarrow$  S  $\leftarrow$  S  $\leftarrow$  S  $\leftarrow$  0
      0      1      2      3      4

o26 : ChainComplex
```

Comparing the multigraded Betti tables of these two resolutions we see that once again the virtual resolution is shorter and less complicated than the graded minimal free resolution.

```
i27 : multigraded betti minres
o27 = 0: 1 . . . .
      5: . a3b2+3a2b3 . . . .
      6: . a4b2+2a2b4+5ab5 . . . .
      7: . . b7 3a4b3+6a3b4+12a2b5+3ab6 . . . .
      8: . . . 2ab7 3a4b4+8a3b5+6a2b6 . . . .
      9: . . . . a2b7 a4b5+3a3b6 . . . .
o27 : MultigradedBettiTally
i28 : multigraded betti vres
o28 = 0: 1 . . . .
      5: . a3b2+3a2b3 . . . .
      6: . 2a2b4+5ab5 . . . .
      7: . . 6a3b4+12a2b5 . . . .
      8: . . . 8a3b5 . . . .
o28 : MultigradedBettiTally
```

**Remark 3.3.** Since the curve generated by `randomCurveP1P2` is, in some sense, random, one will often get different virtual resolutions when running the above example. As an example of this phenomena, working over a slightly larger finite field than in Example 3.1, we compute the multigraded regularity for 500 curves arising from curves of genus 3 and degree 7 in  $\mathbb{P}^3$ . The resulting distribution of multigraded regularities is shown below.

```
i29 : tally apply(500,i->(
      I := randomCurveP1P2(7, 3, ZZ/32003);
      S := ring I;
      B := intersect(ideal(x_(0,0), x_(0,1)),
                    ideal(x_(1,0), x_(1,1), x_(1,2)));
      J := saturate(I, B);
      multigradedRegularity(S, S^1/J)
    ))
o29 = Tally{{{1, 4}, {2, 3}} => 167}
      {{{1, 5}} => 170}
      {{{1, 6}} => 163}
o29 : Tally
```

**Remark 3.4.** It would be interesting to have an understanding of how the geometry of the curves in  $\mathbb{P}^3$  affects multigraded regularity of the resulting curves in  $\mathbb{P}^1 \times \mathbb{P}^2$ . For example, in the computation in Remark 3.3, every smooth curve of genus 3 and degree 7 lies on either a smooth cubic surface or a rational quartic surface with a double line. In fact, using the notation from [Zhang 2018], each curve of genus 3 and degree 7 lies in one of three possible divisor classes:  $(4, 1, 1, 1, 1, 1, 0)$ ,  $(5, 3, 1, 1, 1, 1, 1)$ , and  $(6, 1, 2, 2, 2, 2, 2, 2, 1)$ , the first two being divisors on a smooth cubic surface and the last being a divisor on a rational quartic surface with a double line. An experiment similar to the one in Remark 3.3 suggests the following relationship between divisors and multigraded regularity:

- (1) The minimal elements in the multigraded regularity of any curve in  $\mathbb{P}^1 \times \mathbb{P}^2$  arising from a smooth curve in  $\mathbb{P}^3$  lying on smooth cubic surface in the divisor class  $(4, 1, 1, 1, 1, 1, 0)$  are  $\{(1, 6)\}$ .
- (2) The minimal elements in the multigraded regularity of any curve in  $\mathbb{P}^1 \times \mathbb{P}^2$  arising from a smooth curve in  $\mathbb{P}^3$  lying on smooth cubic surface in the divisor class  $(5, 3, 1, 1, 1, 1, 1)$  are  $\{(1, 5)\}$ .
- (3) The minimal elements in the multigraded regularity of any curve in  $\mathbb{P}^1 \times \mathbb{P}^2$  arising from a smooth curve in  $\mathbb{P}^3$  lying on a rational quartic surfaces with a double line in the divisor class  $(6, 1, 2, 2, 2, 2, 2, 2, 1)$  are  $\{(1, 4), (2, 3)\}$ .

How this generalizes to all curves though remains unclear, but we hope these functions will stimulate others to consider this and other similar problems.

The functions `randomRationalCurve` and `randomMonomialCurve` are alternative methods for generating curves in  $\mathbb{P}^1 \times \mathbb{P}^2$ . Given two positive integers  $d$  and  $e$ , these functions construct the defining ideal of the curve arising as the image of the map

$$\mathbb{P}^1 \rightarrow \mathbb{P}^1 \times \mathbb{P}^2 \quad \text{given by} \quad [t_0 : t_1] \mapsto ([f_0 : f_1], [g_0 : g_1 : g_2]),$$

where the  $f_i$  and  $g_i$  are forms in  $\mathbb{K}[t_0, t_1]$  of degrees  $d$  and  $e$ , respectively. For `randomMonomialCurve`,

these forms are chosen to be monomials, while for `randomRationalCurve` they can be any possible forms of the correct degree.

**Example 3.5.** Here we use the `randomRationalCurve` function to construct a rational curve of bidegree (5, 7). Once again we verify that it is, in fact, a curve by computing its dimension.

```
i30 : I = randomRationalCurve(5, 7);
      ZZ
o30 : Ideal of ---[x , x , y , y , y ]
      101 0 1 0 1 2
i31 : dim I
o31 = 3
```

ACKNOWLEDGMENTS. We would like to thank Christine Berkesch, David Eisenbud, Daniel Erman, Gregory G. Smith, and Mike Stillman for helping to develop the `VirtualResolutions` package. Further, we especially thank Christine Berkesch for reading a draft of this article and providing many helpful comments. We would also like to thank the referees for many helpful comments.

Work on this article and the associated *Macaulay2* package began at the *Macaulay2 Workshop at the University of Wisconsin - Madison* funded by NSF grant DMS-1812462. Additional progress was made during the *Graduate Workshop in Commutative Algebra for Women and Mathematicians of Minority Genders* funded by NSF grant DMS-1908799. The final portion of this article and package was written during a Coding Sprint hosted by the Institute for Mathematics and its Applications (IMA), and we are grateful for their hospitality.

SUPPLEMENT. The online supplement contains version 1.2 of `VirtualResolutions`.

#### REFERENCES.

- [Berkesch et al. 2017] C. Berkesch, D. Erman, and G. G. Smith, “Virtual resolutions for a product of projective spaces”, 2017. to appear in *Alg. Geom.* arXiv
- [Cox 1995] D. A. Cox, “The homogeneous coordinate ring of a toric variety”, *J. Algebraic Geom.* **4**:1 (1995), 17–50. MR Zbl
- [Cox et al. 2011] D. A. Cox, J. B. Little, and H. K. Schenck, *Toric varieties*, Graduate Studies in Mathematics **124**, American Mathematical Society, Providence, RI, 2011. MR Zbl
- [Eisenbud et al. 2015] D. Eisenbud, D. Erman, and F.-O. Schreyer, “Tate resolutions for products of projective spaces”, *Acta Math. Vietnam.* **40**:1 (2015), 5–36. MR Zbl
- [Erman et al. 2019] D. Erman, D. Eisenbud, and F.-O. Schreyer, “Tate Resolutions on Products of Projective Spaces: Cohomology and Direct Image Complexes”, 2019. arXiv
- [Loper 2019] M. C. Loper, “What makes a complex virtual”, 2019. arXiv
- [Macaulay2] D. R. Grayson and M. E. Stillman, “Macaulay2: a software system for research in algebraic geometry”, available at <http://www.math.uiuc.edu/Macaulay2>.
- [Maclagan and Smith 2004] D. Maclagan and G. G. Smith, “Multigraded Castelnuovo–Mumford regularity”, *J. Reine Angew. Math.* **571** (2004), 179–212. MR Zbl
- [Zhang 2018] M. Zhang, “The SpaceCurves package in Macaulay2”, *J. Softw. Algebra Geom.* **8** (2018), 31–48. MR

AYAH ALMOUSA:

aka66@cornell.edu

Department of Mathematics, Cornell University, Ithaca, NY, United States

JULIETTE BRUCE:

juliette.bruce@math.wisc.edu

Department of Mathematics, University of Wisconsin-Madison, Madison, WI, United States

MICHAEL LOPER:

loper012@umn.edu

School of Mathematics, University of Minnesota, Minneapolis, MN, United States

MAHRUD SAYRAFI:

mahrud@umn.edu

School of Mathematics, University of Minnesota, Minneapolis, MN, United States